Chapter 9

AC Sweep and Signal Analysis

This chapter describes performing an AC sweep and small signal analysis. It covers the following topics:

- Understanding AC Small Signal Analysis
- Using the .AC Statement
- Using Other AC Analysis Statements

Understanding AC Small Signal Analysis

The AC small signal analysis portion of Star-Hspice computes (see Figure 9-1) AC output variables as a function of frequency. Star-Hspice first solves for the DC operating point conditions, which are used to develop linearized, small-signal models for all nonlinear devices in the circuit.



Figure 9-1: AC Small Signal Analysis Flow

Capacitor and inductor values are converted to their corresponding admittances:

$Y_C = j\omega C$ for capacitors

and

$Y_L = 1/j\omega L$ for inductors

Star-Hspice allows resistors to have different DC and AC values. If AC=<value> is specified in a resistor statement, the operating point is calculated using the DC value of resistance, but the AC resistance value is used in the AC analysis. This is convenient when analyzing operational amplifiers, since the operating point computation can be performed on the unity gain configuration using a low value for the feedback resistance. The AC analysis then can be performed on the open loop configuration by using a very large value for the AC resistance.

AC analysis of bipolar transistors is based on the small-signal equivalent circuit, as described in “Using the BJT Models (NPN and PNP)” on page 14-33. MOSFET AC equivalent circuit models are described in *Chapter , Introducing MOSFET*.

The AC analysis statement permits sweeping values for:

- Frequency
- Element
- Temperature
- Model parameter
- Randomized distribution (Monte Carlo)
- Optimization and AC design analysis

Additionally, as part of the small signal analysis tools, Star-Hspice provides:

- Noise analysis
- Distortion analysis
- Network analysis
- Sampling noise

Using the .AC Statement

You can use the .AC statement in several different formats, depending on the application, as shown in the examples below. The parameters are described below.

Syntax

Single/double sweep:

```
.AC type np fstart fstop
```

or

```
.AC type np fstart fstop <SWEEP var starstop incr>
```

or

```
.AC type np fstart fstop <SWEEP var type np start stop>
```

or

```
.AC var1 START= <param_expr1> STOP= <param_expr2>
+  STEP = <param_expr3>
```

or

```
.AC var1 START = start1 STOP = stop1 STEP = incr1
```

Parameterized sweep:

```
.AC type np fstart fstop <SWEEP DATA=datanm>
```

or

```
.AC DATA=datanm
```

Optimization:

```
.AC DATA=datanm OPTIMIZE=opt_par_fun RESULTS=measnames
+  MODEL=optmod
```

Random/Monte Carlo:

```
.AC type np fstart fstop <SWEEP MONTE=val>
```

The .AC statement keywords and parameters have the following descriptions:

DATA=datanm data name referred to in the .AC statement

incr	voltage, current, element or model parameter increment value **Note:** If “type” variation is used, the “np” (number of points) is specified instead of “incr”.
fstart	starting frequency **Note:** If type variation “POI” (list of points) is used, a list of frequency values is specified instead of “fstart fstop”.
fstop	final frequency
MONTE=val	produces a number *val* of randomly-generated values that are used to select parameters from a distribution. The distribution can be *Gaussian*, *Uniform*, or *Random Limit*. See “Performing Monte Carlo Analysis” on page 10-39 for more information.
np	number of points per decade or per octave, or just number of points, depending on the preceding keyword
start	starting voltage, current, any element or model parameter value
stop	final voltage, current, any element or model parameter value
SWEEP	keyword to indicate a second sweep is specified in the .AC statement
TEMP	keyword to indicate a temperature sweep
type	can be any of the following keywords: DEC – decade variation OCT – octave variation LIN – linear variation POI – list of points
var	name of an independent voltage or current source, any element or model parameter, or the keyword TEMP (indicating a temperature sweep). Star-Hspice supports

source value sweep, referring to the source name (SPICE style). However, if parameter sweep, a .DATA statement, and temperature sweep are selected, a parameter name must be chosen for the source value and subsequently referred to in the .AC statement. The parameter name can not start with V or I.

Examples

The following example performs a frequency sweep by 10 points per decade from 1 kHz to 100 MHz.

```
.AC DEC  10  1K 100MEG
```

The next line calls for a 100 point frequency sweep from 1 Hz to 100 Hz.

```
.AC LIN  100 1  100HZ
```

The following example performs an AC analysis for each value of cload, which results from a linear sweep of cload between 1 pF and 10 pF (20 points), sweeping frequency by 10 points per decade from 1 Hz to 10 kHz.

```
.AC DEC  10  1  10K  SWEEP  cload  LIN  20  1pf  10pf
```

The following example performs an AC analysis for each value of rx, 5 k and 15 k, sweeping frequency by 10 points per decade from 1 Hz to 10 kHz.

```
.AC DEC  10  1  10K  SWEEP  rx n POI  2  5k  15k
```

The next example uses the DATA statement to perform a series of AC analyses modifying more than one parameter. The parameters are contained in the file *datanm*.

```
.AC DEC  10  1  10K  SWEEP  DATA=datanm
```

The following example illustrates a frequency sweep along with a Monte Carlo analysis with 30 trials.

```
.AC DEC  10  1  10K  SWEEP  MONTE=30
```

When an .AC statement is included in the input file, Star-Hspice performs an AC analysis of the circuit over the specified frequency range for each parameter value specified in the second sweep.

For an AC analysis, at least one independent AC source element statement must be in the data file (for example, VI INPUT GND AC 1V). Star-Hspice checks for this condition and reports a fatal error if no such AC sources have been specified (see Chapter , Using Sources and Stimuli).

AC Control Options

ABSH=x Sets the absolute current change through voltage defined branches (voltage sources and inductors). In conjunction with DI and RELH, ABSH is used to check for current convergence. Default=0.0.

ACOUT AC output calculation method for the difference in values of magnitude, phase and decibels for prints and plots. Default=1.

The default value, ACOUT=1, selects the Star-Hspice method, which calculates the difference of the magnitudes of the values. The SPICE method, ACOUT=0, calculates the magnitude of the differences.

DI=x Sets the maximum iteration-to-iteration current change through voltage defined branches (voltage sources and inductors). This option is only applicable when the value of the DI control option is greater than 0. Default=0.0.

MAXAMP=x Sets the maximum current through voltage defined branches (voltage sources and inductors). If the current exceeds the MAXAMP value, an error message is issued. Default=0.0.

RELH=x Sets relative current tolerance through voltage defined branches (voltage sources and inductors). It is used to check current convergence. This option is applicable only if the value of the ABSH control option is greater than zero. Default=0.05.

UNWRAP displays phase results in AC analysis in unwrapped form (with a continuous phase plot). This allows accurate calculation of group delay. Note that group delay is always computed based on unwrapped phase results, even if the UNWRAP option is not set.

AC Analysis Output Variables

Output variables for AC analysis include:

- Voltage differences between specified nodes (or one specified node and ground)
- Current output for an independent voltage source
- Element branch current
- Impedance (Z), admittance (Y), hybrid (H), and scattering (S) parameters
- Input and output impedance and admittance

AC output variable types are listed in Table 9-1:. The type symbol is appended to the variable symbol to form the output variable name. For example, VI is the imaginary part of the voltage, or IM is the magnitude of the current.

Table 9-1: AC Output Variable Types.

Type Symbol	Variable Type
DB	decibel
I	imaginary part
M	magnitude
P	phase
R	real part
T	group delay

Specify real or imaginary parts, magnitude, phase, decibels, and group delay for voltages and currents.

AC Nodal Voltage Output

Syntax

```
Vx (n1,<,n2>)
```

where:

x	specifies the voltage output type (see Table 9-1:)
n1, n2	specfies node names. If n2 is omitted, ground (node 0) is assumed.

Example

```
.PLOT AC VM(5) VDB(5) VP(5)
```

The above example plots the magnitude of the AC voltage of node 5 using the output variable VM. The voltage at node 5 is plotted with the VDB output variable. The phase of the nodal voltage at node 5 is plotted with the VP output variable.

Since an AC analysis produces complex results, the values of real or imaginary parts of complex voltages of AC analysis and their magnitude, phase, decibel, and group delay values are calculated using either the SPICE or Star-Hspice method and the control option ACOUT. The default for Star-Hspice is ACOUT=1. To use the SPICE method, set ACOUT=0.

The SPICE method is typically used to calculate the nodal vector difference in comparing adjacent nodes in a circuit. It is used to find phase or magnitude across a capacitor, inductor, or semiconductor device.

Use the Star-Hspice method to calculate an interstage gain in a circuit (such as an amplifier circuit) and to compare its gain, phase, and magnitude.

The following examples define the AC analysis output variables for the Star-Hspice and then for the SPICE method.

Star-Hspice Method (ACOUT=1, Default)

Real and imaginary:

$$VR(N1,N2) = REAL\ [V(N1,0)] - REAL\ [V(N2,0)]$$

$$VI(N1,N2) = IMAG\ [V(N1,0)] - IMAG\ [V(N2,0)]$$

Magnitude:

$$VM(N1,0) = [VR(N1,0)^2 + VI(N1,0)^2]^{0.5}$$

$$VM(N2,0) = [VR(N2,0)^2 + VI(N2,0)^2]^{0.5}$$

$$VM(N1,N2) = VM(N1,0) - VM(N2,0)$$

Phase:

$$VP(N1,0) = ARCTAN[VI(N1,0)/VR(N1,0)]$$

$$VP(N2,0) = ARCTAN[VI(N2,0)/VR(N2,0)]$$

$$VP(N1,N2) = VP(N1,0) - VP(N2,0)$$

Decibel:

$$VDB(N1,N2) = 20 \cdot LOG10(VM(N1,0)/VM(N2,0))$$

SPICE Method (ACOUT=0)

Real and imaginary:

$$VR(N1,N2) = REAL\ [V(N1,0) - V(N2,0)]$$

$$VI(N1,N2) = IMAG\ [V(N1,0) - V(N2,0)]$$

Magnitude:

$$VM(N1,N2) = [VR(N1,N2)^2 + VI(N1,N2)^2]^{0.5}$$

Phase:

$$VP(N1,N2) = ARCTAN[VI(N1,N2)/VR(N1,N2)]$$

Decibel:

$$VDB(N1,N2) = 20 \cdot LOG10[VM(N1,N2)]$$

AC Current Output: Independent Voltage Sources

Syntax

```
Iz (Vxxx)
```

where:

z the current output type (see Table 9-1:)

Vxxx voltage source element name. If an independent power supply is within a subcircuit, its current output is accessed by appending a dot and the subcircuit name to the element name, for example, IM(X1.Vyyy).

Example

```
.PLOT AC IR(V1) IM(VN2B) IP(X1.X2.VSRC)
```

AC Current Output: Element Branches

Syntax

```
Izn (Wwww)
```

where:

z current output type (see Table 9-1:)

n node position number in the element statement. For example, if the element contains four nodes, IM3 denotes the magnitude of the branch current output for the third node.

Wwww element name. If the element is within a subcircuit, its current output is accessed by appending a dot and the subcircuit name to the element name, for example, IM3(X1.Qyyy).

Example

```
.PRINT AC IP1(Q5) IM1(Q5) IDB4(X1.M1)
```

If the form In(Xxxx) is used for AC analysis output, the magnitude IMn(Xxxx) is the value printed.

Group Time Delay Output

Syntax

```
.PRINT AC VT(10) VT(2,25) IT(RL)
.PLOT AC IT1(Q1) IT3(M15) IT(D1)
```

Note: Since there is discontinuity in phase each 360°, the same discontinuity is seen in TD, even though TD is continuous.

Example

```
INTEG.SP ACTIVE INTEGRATOR
******  INPUT LISTING
******
V1       1   0  .5   AC   1
R1       1   2      2K
C1       2   3      5NF
E3       3   0      2 0 -1000.0
.AC DEC   15 1K  100K
.PLOT AC VT(3) (0,4U) VP(3)
.END
```

AC Network Output

Syntax

```
Xij (z), ZIN(z), ZOUT(z), YIN(z), YOUT(z)
```

where

X specifies Z for impedance, Y for admittance, H for hybrid, or S for scattering parameters

ij	i and j can be 1 or 2. They identify which matrix parameter is printed.
z	output type (see Table 9-1:). If z is omitted, the magnitude of the output variable is printed.
ZIN	input impedance. For a one port network ZIN, Z11, and H11 are the same
ZOUT	output impedance
YIN	input admittance. For a one-port network, YIN and Y11 are the same.
YOUT	output admittance

Examples

```
.PRINT AC Z11(R) Z12(R) Y21(I) Y22 S11 S11(DB)
.PRINT AC ZIN(R) ZIN(I) YOUT(M) YOUT(P) H11(M)
.PLOT  AC S22(M) S22(P) S21(R)  H21(P) H12(R)
```

Using Other AC Analysis Statements

This section describes how to use other AC analysis statements.

.DISTO Statement — AC Small-Signal Distortion Analysis

The .DISTO statement causes Star-Hspice to compute the distortion characteristics of the circuit in an AC small-signal, sinusoidal, steady-state analysis.The program computes and reports five distortion measures at the specified load resistor. The analysis is performed assuming that one or two signal frequencies are imposed at the input. The first frequency, F1 (used to calculate harmonic distortion), is the nominal analysis frequency set by the .AC statement frequency sweep. The optional second input frequency, F2 (used to calculate intermodulation distortion), is set implicitly by specifying the parameter skw2, which is the ratio F2/F1.

DIM2	Intermodulation distortion, difference. The relative magnitude and phase of the frequency component (F1 - F2).
DIM3	Intermodulation distortion, second difference. The relative magnitude and phase of the frequency component (2 · F1 - F2).
HD2	Second order harmonic distortion. The relative magnitude and phase of the frequency component 2 · F1 (ignoring F2).
HD3	Third order harmonic distortion. The relative magnitude and phase of the frequency component 3 · F1 (ignoring F2).
SIM2	Intermodulation distortion, sum. The relative magnitude and phase of the frequency component (F1 + F2).

The .DISTO summary report includes a set of distortion measures for each contributing component of every element, a summary set for each element, and a set of distortion measures representing a sum over all the elements in the circuit.

Syntax

```
.DISTO Rload <inter <skw2 <refpwr <spwf>>>>
```

where:

Rload	the resistor element name of the output load resistor into which the output power is fed
inter	interval at which a distortion-measure summary is to be printed. Specifies a number of frequency points in the AC sweep (see the np parameter in “Using the .AC Statement”). If inter is omitted or set to zero, no summary printout is made. In this case, the distortion measures can be printed or plotted with the .PRINT or .PLOT statement. If inter is set to 1 or higher, a summary printout is made for the first frequency, and once for each inter frequency increment thereafter. To obtain a summary printout for only the first and last frequencies, set inter equal to the total number of increments needed to reach fstop in the .AC statement. For a summary printout of only the first frequency, set inter to greater than the total number of increments required to reach fstop.
skw2	ratio of the second frequency F2 to the nominal analysis frequency F1. The acceptable range is 1e-3 < skw2 ≤ 0.999. If skw2 is omitted, a value of 0.9 is assumed.
refpwr	reference power level used in computing the distortion products. If omitted, a value of 1mW, measured in decibels magnitude (dbM), is assumed. The value must be ≥ 1e-10.
spwf	amplitude of the second frequency F2. The value must be ≥ 1e-3. Default=1.0.

Example

```
.DISTO RL 2 0.95 1.0E-3 0.75
```

Only one distortion analysis is performed per simulation. If more than one .DISTO statement is found, only the last is performed.

Note: The summary printout from the distortion analysis for each frequency listed is extensive. Use the “inter” parameter in the .DISTO statement to limit the amount of output generated.

.NOISE Statement — AC Noise Analysis

Syntax

```
.NOISE ovv srcnam inter
```

where:

ovv	nodal voltage output variable defining the node at which the noise is summed
srcnam	name of the independent voltage or current source to be used as the noise input reference
inter	interval at which a noise analysis summary is to be printed, inter specifies a number of frequency points summary in the AC sweep. If inter is omitted or set to zero, no summary printout is made. If inter is equal to or greater than one, a summary printout is made for the first frequency, and once for each inter frequency increment thereafter.

Example

```
.NOISE V(5) VIN 10
```

The .NOISE statement, used in conjunction with the AC statement, controls the noise analysis of the circuit.

Noise Calculations

The noise calculations in Star-Hspice are based on the complex AC nodal voltages, which in turn are based on the DC operating point. Noise models are described for each device type in the appropriate chapter in Volume II. A noise source is not assumed to be statistically correlated to the other noise sources in the circuit; each noise source is calculated independently. The total output noise voltage is the RMS sum of the individual noise contributions:

$$onoise = \sum_{n=1}^{n} \left| Z_n \cdot I_n \right|^2$$

where:

onoise	total output noise
I	equivalent current due to thermal noise, shot or flicker noise
Z	equivalent transimpedance between noise source and the output
n	number of noise sources associated with all resistors, MOSFETs, diodes, JFETs, and BJTs

The equivalent input noise voltage is the total output noise divided by the gain or transfer function of the circuit. The contribution of each noise generator in the circuit is printed for each inter frequency point. The output and input noise levels are normalized with respect to the square root of the noise bandwidth, and have the units volts/Hz$^{1/2}$ or amps/Hz$^{1/2}$.

You can simulate flicker noise sources in the noise analysis by including values for the parameters KF and AF on the appropriate device model statements.

Use the .PRINT or .PLOT statement to print or plot the output noise and the equivalent input noise.

You can only perform one noise analysis per simulation. If more than one NOISE statement is present, only the last one is performed.

.SAMPLE Statement — Noise Folding Analysis

For data acquisition of analog signals, data sampling noise often needs to be analyzed. This is accomplished with the .SAMPLE statement used in conjunction with the .NOISE and .AC statements.

The SAMPLE analysis causes Star-Hspice to perform a simple noise folding analysis at the output node.

Syntax

```
.SAMPLE FS=freq <TOL=val> <NUMF=val> <MAXFLD=val> <BETA=val>
```

where:

FS=freq sample frequency, in Hertz

TOL sampling error tolerance: the ratio of the noise power in the highest folding interval to the noise power in baseband. Default=1.0e-3.

NUMF maximum allowed number of user-specified frequencies. The algorithm requires approximately ten times this number of internally generated frequencies, so it should be kept small. Default=100.

MAXFLD maximum allowed number of folding intervals. The highest frequency (in Hertz) considered by the algorithm is given by:

$$FMAX = MAXFLD \cdot FS$$

Default=10.0.

BETA Integrator duty cycle; specifies an optional noise integrator at the sampling node

BETA=0 no integrator

BETA=1 simple integrator (default)

If the integrator is clocked (that is, it only integrates during a fraction of the sampling interval 1/FS), then BETA should be set to the duty cycle of the integrator.

.NET Statement - AC Network Analysis

The .NET statement computes the parameters for the impedance matrix Z, the admittance matrix Y, the hybrid matrix H, and the scattering matrix S. The input impedance, output impedance, and admittance are also computed. This analysis is a part of the AC small-signal analysis. Therefore, network analysis requires the specification of the AC statement frequency sweep.

Syntax

One-port network:

```
.NET input <RIN=val>
```

or

```
.NET input <val >
```

Two-port network:

```
.NET output input <ROUT=val> <RIN=val>
```

where:

input	AC input voltage or current source name
output	output port. It can be an output voltage, V(n1,n2), or an output current, I(source), or I(element).
RIN	input or source resistance keyword. The RIN value is used to calculate the output impedance and admittance, and also the scattering parameters. The RIN value defaults to 1 ohm.
ROUT	output or load resistance keyword. The ROUT value is used to calculate the input impedance and admittance, and also the scattering parameters. The ROUT value defaults to 1 ohm.

Examples

One-port network:

```
.NET    VINAC     RIN=50
.NET    IIN       RIN=50
```

Two-port network:

```
.NET  V(10,30)  VINAC   ROUT=75   RIN=50
.NET  I(RX)     VINAC   ROUT=75   RIN=50
```

AC Network Analysis - Output Specification

Syntax

```
Xij(z), ZIN(z), ZOUT(z), YIN(z), YOUT(z)
```

where:

X	specifies Z for impedance, Y for admittance, H for hybrid, and S for scattering
ij	i and j can be 1 or 2. They identify which matrix parameter is to be printed.
z	output type: R: real part I : maginary part M: magnitude P: phase DB: decibel T: group time delay
ZIN	input impedance. For the one port network, ZIN, Z11 and H11 are the same.
ZOUT	output impedance
YIN	input admittance. For the one port network, YIN and Y11 are the same.
YOUT	output admittance

If “z” is omitted, output includes the magnitude of the output variable.

Examples

```
.PRINT  AC Z11(R)  Z12(R)  Y21(I)  Y22 S11 S11(DB) Z11(T)
.PRINT  AC ZIN(R) ZIN(I) YOUT(M) YOUT(P) H11(M) H11(T)
.PLOT   AC S22(M) S22(P) S21(R) H21(P) H12(R) S22(T)
```

Bandpass Netlist:[1] Star-Hspice Network Analysis Results

```
*FILE: FBP_1.SP
.OPTIONS DCSTEP=1 POST
*BAND PASS FILTER
C1  IN   2   3.166PF
L1  2    3   203NH
C2  3    0   3.76PF
C3  3    4   1.75PF
C4  4    0   9.1PF
L2  4    0  36.81NH
C5  4    5   1.07PF
C6  5    0   3.13PF
L3  5    6 233.17NH
C7  6    7   5.92PF
C8  7    0   4.51PF
C9  7    8   1.568PF
C10 8    0   8.866PF
L4  8    0  35.71NH
C11 8    9   2.06PF
C12 9    0   4.3PF
L5  9   10 200.97NH
C13 10 OUT   2.97PF
RX  OUT  0   1E14
VIN IN   0   AC 1
.AC LIN 41 200MEG 300MEG
.NET V(OUT) VIN ROUT=50 RIN=50
.PLOT AC S11(DB) (-50,10) S11(P) (-180,180)
.PLOT AC ZIN(M) (5,130) ZIN(P) (-90,90)
.END
```



Figure 9-2: S11 Magnitude and Phase Plots



Figure 9-3: ZIN Magnitude and Phase Plots

NETWORK Variable Specification

Star-Hspice uses the results of AC analysis to perform network analysis. The .NET statement defines the Z, Y, H, and S parameters to be calculated.

The following list shows various combinations of the .NET statement for network matrices that are initially calculated in Star-Hspice:

1)	.NET Vout Isrc	V	=	[Z]	[I]
2)	.NET Iout Vsrc	I	=	[Y]	[V]
3)	.NET Iout Isrc	$[V1\ I2]^T$	=	[H]	$[I1\ V2]^T$
4)	.NET Vout Vsrc	$[I1\ V2]^T$	=	[S]	$[V1\ I2]^T$

($[M\ ]^T$ represents the *transpose* of matrix M)

Note: The preceding list does not mean that combination (1) must be used for calculating the Z parameters. However, if .NET Vout Isrc is specified, Star-Hspice initially evaluates the Z matrix parameters and then uses standard conversion equations to determine the S parameters or any other requested parameters.

The example in Figure 9-4: shows the importance of the variables used in the .NET statement. Here, *Isrc* and *Vce* are the DC biases applied to the BJT.



Figure 9-4: Parameters with .NET V(2) Isrc

This .NET statement provides an incorrect result for the Z parameters calculation:

```
.NET V(2) Isrc
```

When Star-Hspice performs AC analysis, all the DC voltage sources are shorted and all the DC current sources are open-circuited. As a result, V(2) is shorted to ground, and its value is zero for AC analysis, directly affecting the results of the network analysis. When Star-Hspice attempts to calculate the Z parameters Z11 and Z21, defined as Z11 = V1/I1 and Z21 = V2/I1 with I2=0, the requirement that I2 must be zero is not satisfied in the circuit above. Instead, V2 is zero, which results in incorrect values for Z11 and Z21.

The correct biasing configurations for performing network analysis for Z, Y, H, and S parameters are shown in Figure 9-5:.



Figure 9-5: Network Parameter Configurations

As an example, the H parameters are calculated by using the .NET statement.

```
.NET I(VC) IB
```

Here, V_C denotes the voltage at node C, the collector of the BJT. With this statement, Star-Hspice calculates the H parameters immediately after AC analysis. The H parameters are calculated by:

$$V1 = H11 \cdot I1 + H12 \cdot V2$$

$$I2 = H21 \cdot I1 + H22 \cdot V2$$

For Hybrid parameter calculations of H11 and H21, V2 is set to zero (due to the DC voltage source V_{CE}), while for H12 and H22 calculations, I1 is set to zero (due to the DC current source I_B). Setting I1 and V2 equal to zero precisely meets the conditions of the circuit under examination; namely, that the input current source is open circuited and the output voltage source is shorted to ground.

External DC biases applied to a BJT can be driven by a data file of measured results. In some cases, not all of the DC currents and voltages at input and output ports are available. When performing network analysis, examine the circuit and select suitable input and output variables to obtain correctly calculated results. The following examples demonstrate the network analysis of a BJT using Star-Hspice.

Network Analysis Example: Bipolar Transistor

```
BJT network analysis
.option  nopage list
+        newtol reli=1e-5 absi=1e-10 relv=1e-5 relvdc=1e-7
+        nomod post gmindc=1e-12
.op
.param vbe=0 ib=0 ic=0 vce=0

$ H-parameter
.NET i(vc) ibb rin=50 rout=50
ve    e   0       0
ibb   0   b       dc='ib' ac=0.1
vc    c   0       'vce'
q1    c   b  e 0   bjt

.model bjt  npn  subs=1
+ bf=1.292755e+02 br=8.379600e+00
+ is=8.753000e-18 nf=9.710631e-01 nr=9.643484e-01
+ ise=3.428000e-16 isc=1.855000e-17 iss=0.000000e+00
```

```
+ ne=2.000000e+00 nc=9.460594e-01 ns=1.000000e+00
+ vaf=4.942130e+01 var=4.589800e+00
+ ikf=5.763400e-03 ikr=5.000000e-03 irb=8.002451e-07
+ rc=1.216835e+02 rb=1.786930e+04 rbm=8.123460e+01
+ re=2.136400e+00
+ cje=9.894950e-14 mje=4.567345e-01 vje=1.090217e+00
+ cjc=5.248670e-14 mjc=1.318637e-01 vjc=5.184017e-01
+ xcjc=6.720303e-01
+ cjs=9.671580e-14 mjs=2.395731e-01 vjs=5.000000e-01
+ tf=3.319200e-11 itf=1.457110e-02 xtf=2.778660e+01
+ vtf=1.157900e+00 ptf=6.000000e-05
+ xti=4.460500e+00 xtb=1.456600e+00 eg=1.153300e+00
+ tikf1=-5.397800e-03 tirb1=-1.071400e-03
+ tre1=-1.121900e-02 trb1=3.039900e-03
+ trc1=-4.020700e-03 trm1=0.000000e+00

.print ac par('ib') par('ic')
+ h11(m) h12(m) h21(m) h22(m)
+ z11(m) z12(m) z21(m) z22(m)
+ s11(m) s21(m) s12(m) s22(m)
+ y11(m) y21(m) y12(m) y22(m)

.ac Dec 10  1e6  5g  sweep  data=bias


.data bias
     vbe          vce            ib            ic
 771.5648m    292.5047m      1.2330u    126.9400u
 797.2571m    323.9037m      2.6525u    265.0100u
 821.3907m    848.7848m      5.0275u    486.9900u
 843.5569m      1.6596       8.4783u    789.9700u
 864.2217m      2.4031      13.0750u      1.1616m
 884.3707m      2.0850      19.0950u      1.5675m
.enddata
.end
```

Other possible biasing configurations for the network analysis follow.

$S-parameter

```
.NET v(c) vbb rin=50 rout=50
ve   e 0      0
vbb  b 0      dc='vbe' ac=0.1
icc  0 c      'ic'
q1   c b e 0  bjt
```

$Z-parameter

```
.NET v(c) ibb rin=50 rout=50
ve  e 0       0
ibb  0 b      dc='ib' ac=0.1
icc  0 c      'ic'
q1  c b e 0   bjt
```

$Y-parameter

```
.NET i(vc) vbb rin=50 rout=50
ve  e 0      0
vbb b 0      'vbe' ac=0.1
vc  c 0      'vce'
q1  c b e 0  bjt
```

References

1. Goyal, Ravender. “S-Parameter Output From SPICE Program”, *MSN & CT*, February 1988, pp. 63 and 66.